ΑΣΚΗΣΕΙΣ ΣΕ ΣΤΟΙΒΑ ΚΑΙ ΟΥΡΑ

1. Να ωθηθούν 5 αριθμοί σε στοίβα και στη συνέχεια να απωθηθούν και να εμφανιστούν με τη σειρά που απωθούνται. 

ΠΡΟΓΡΑΜΜΑ ΠΕΝΤΕ_ΑΡΙΘΜΟΙ_ΣΕ_ΣΤΟΙΒΑ
ΜΕΤΑΒΛΗΤΕΣ
	ΑΚΕΡΑΙΕΣ: Stack[100], i, top, number, item
	ΛΟΓΙΚΕΣ: done
ΑΡΧΗ
	top <-- 0
	ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 5
		ΔΙΑΒΑΣΕ number
		ΚΑΛΕΣΕ ΩΘΗΣΗ (Stack, number, top, done)
	ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
	ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 5
		ΚΑΛΕΣΕ ΑΠΩΘΗΣΗ (Stack, item, top, done)
		ΓΡΑΨΕ item
	ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

ΔΙΑΔΙΚΑΣΙΑ ΩΘΗΣΗ (Stack, item, top, done)
ΜΕΤΑΒΛΗΤΕΣ
	ΑΚΕΡΑΙΕΣ: Stack[100], item, top
	ΛΟΓΙΚΕΣ: done
ΑΡΧΗ
	ΑΝ top < 100 ΤΟΤΕ
		top <-- top + 1
		Stack[top] <-- item
		done <-- ΑΛΗΘΗΣ
	ΑΛΛΙΩΣ
		done <-- ΨΕΥΔΗΣ
	ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ

ΔΙΑΔΙΚΑΣΙΑ ΑΠΩΘΗΣΗ (Stack, item, top, done)
ΜΕΤΑΒΛΗΤΕΣ
	ΑΚΕΡΑΙΕΣ: Stack[100], item, top
	ΛΟΓΙΚΕΣ: done
ΑΡΧΗ
	ΑΝ top <> 0 ΤΟΤΕ 	! ή top > = 1
		item <-- Stack[top]
		top <-- top - 1
		done <-- ΑΛΗΘΗΣ
	ΑΛΛΙΩΣ
		done <-- ΨΕΥΔΗΣ
	ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ
2. α) Να γράψετε πρόγραμμα που θα διαβάζει 10 ακεραίους αριθμούς και θα κάνει εισαγωγή των αριθμών σε μια ουρά 10 θέσεων, στη συνέχεια να εξαχθούν 6 αριθμοί οι οποίοι θα εμφανίζονται με τη σειρά που εξάγονται και να εισαχθούν άλλοι 3 ακέραιοι αριθμοί. 
Να εμφανιστούν κατά σειρά οι αριθμοί που βρίσκονται στο τέλος μέσα στην ουρά.
β) Το πρόγραμμα θα καλεί τις διαδικασίες ΕΙΣΑΓΩΓΗ και ΕΞΑΓΩΓΗ που θα υλοποιούν τις αντίστοιχες λειτουργίες της ουράς τις οποίες και θα γράψετε. Αν η ουρά γεμίσει αλλά έχει άδειες θέσεις μπροστά, τότε να μεταφέρονται τα στοιχεία που βρίσκονται στην ουρά μπροστά ώστε να συνεχίζεται η διαδικασία της εισαγωγής.
[bookmark: _GoBack]Θα επιστρέφονται από τις διαδικασίες οι λογικές τιμές ΑΛΗΘΗΣ ή ΨΕΥΔΗΣ αναλόγως αν η εισαγωγή ή η εξαγωγή στοιχείου στην ουρά έγιναν επιτυχώς ή όχι.

ΠΡΟΓΡΑΜΜΑ ΚΥΚΛΙΚΗ_ΟΥΡΑ
ΜΕΤΑΒΛΗΤΕΣ
	ΑΚΕΡΑΙΕΣ: front, rear, i, number, item, Queue[10]
	ΛΟΓΙΚΕΣ: done
ΑΡΧΗ
	front <-- 0
	rear <-- 0
	ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10
		ΓΡΑΨΕ ‘Δώστε ακέραιο αριθμό’
		ΔΙΑΒΑΣΕ number
		ΚΑΛΕΣΕ ΕΙΣΑΓΩΓΗ (Queue, number, front, rear, done)
	ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
	ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 6
		ΚΑΛΕΣΕ ΕΞΑΓΩΓΗ (Queue, item, front, rear, done)
		ΓΡΑΨΕ item
	ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ	
	ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 3
		ΓΡΑΨΕ ‘Δώστε ακέραιο αριθμό’
		ΔΙΑΒΑΣΕ number
		ΚΑΛΕΣΕ ΕΙΣΑΓΩΓΗ (Queue, number, front, rear, done)
	ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
	ΓΙΑ i ΑΠΟ front ΜΕΧΡΙ rear
		ΓΡΑΨΕ Queue[i]
	ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ



ΔΙΑΔΙΚΑΣΙΑ ΕΙΣΑΓΩΓΗ (Queue, item, front, rear, done)
ΜΕΤΑΒΛΗΤΕΣ
	ΑΚΕΡΑΙΕΣ: Queue[10], item, front, rear, S, i
	ΛΟΓΙΚΕΣ: done
ΑΡΧΗ
	ΑΝ rear < 10 ΤΟΤΕ
		rear <-- rear + 1
		Queue[rear] <-- item
		ΑΝ front = 0 ΤΟΤΕ
			front <-- 1  ! Σχόλιο 1
		ΤΕΛΟΣ_ΑΝ
		done <-- ΑΛΗΘΗΣ
	ΑΛΛΙΩΣ_ΑΝ  front = 1 ΤΟΤΕ
		done <-- ΨΕΥΔΗΣ  ! Σχόλιο 2
	ΑΛΛΙΩΣ 		! Σχόλιο 3
		S <-- rear - front + 1 	! Σχόλιο 4	
		ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ S 	
			Queue[i] <-- Queue[front]
			front <-- front + 1
		ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
		front <-- 1
		rear <-- S + 1 	! ή rear <-- i
		Queue[rear] <-- item
		done <-- ΑΛΗΘΗΣ
	ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ

ΔΙΑΔΙΚΑΣΙΑ ΕΞΑΓΩΓΗ (Queue, item, front, rear, done)
ΜΕΤΑΒΛΗΤΕΣ
	ΑΚΕΡΑΙΕΣ: Queue[10], item, front, rear
	ΛΟΓΙΚΕΣ: done
ΑΡΧΗ
	ΑΝ front <= rear ΚΑΙ front <> 0 ΤΟΤΕ
		item <-- Queue[front]
		front <-- front + 1
		done <-- ΑΛΗΘΗΣ
	ΑΛΛΙΩΣ
		done <-- ΨΕΥΔΗΣ
	ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ

Σχόλια:
Σχόλιο 1: Ο δείκτης front πρέπει να γίνει 1 μόλις εισαχθεί το πρώτο στοιχείο.
Σχόλιο 2: Η ουρά είναι γεμάτη από την πρώτη θέση μέχρι τέλος και δε γίνεται άλλη εισαγωγή.
Σχόλιο 3: Δε γίνεται εισαγωγή στο πίσω μέρος της ουράς, αλλά μπροστά η ουρά έχει άδειες  θέσεις, οπότε μεταφέρω τα στοιχεία μπροστά ώστε να γίνει η εισαγωγή.
Σχόλιο 4: Το S είναι το πλήθος των στοιχείων που θα μεταφερθούν.

